Hands-on
OpenModelica
ter supply system model for shallow well pum





2019/12/14 16th Modellca Library Japanese Stud
Session finback

OpenModelica Well pump water supply system model (2)

We will model a pressure tank type shallow well pump and prototype a model to pump well water and supply it.

- Pump
- Stop valve (faucet)
- Pressure tank
- Pressure switch (pump on / off control device)
- Piping system (well, pipe)

They are modeled and combined.

In the second session, we model well, pressure tank, and pressure switch, and build an overall model.

Reference:

Modelica Fluid Library Documentation
https://www.amane.to/archives/285

Modelica Standard Library Modeling of SingleGasNasa Pure Material Ideal Gas in Japanese
https://www.amane.to/archives/114

Today's program

Shallow well

- WellPumpTest3 Shallow well + pump + open tank + valve

Pressure tank

- Modeling air in a pressure tank
- Pressure tank inlet / outlet pressure
- PressureTank Pressure tank model
- PressureTankTest1 Unit test
- PressureTankTest2 Shallow well + pump + pressure tank + valve

Pressure switch

- PressureSwitch

Conclusion

Shallow well

Add a pipe for pumping water from a shallow well to WellPumpTest2 created last time.



- The height of the pump inlet will be 0.3 m from the ground, considering that the pump will be placed on the pressure tank later.
- Pump 6.3 m from the well surface to the pump inlet. Model this part with StaticPipe.
- At the pump inlet, place a ClosedVolume so that the initial pressure can be set.
- Height difference between pump outlet and open tank inlet 0.22 m

WellPumpTest3

① Duplicate WellPumpTest2 to create WellPumpTest3 and modify as follows.

ClosedVolume (Add new)
V = 0.001 [m3]
nPorts = 2
use_portsData = false
energyDynamics = FixedInitial
massDynamics = FixedInitial
p_start = system.p_start - 1000 * 6.3 * Modelica.Constants.g_n

Pump inlet initial pressure

height = 1
duration = 180 [s]
offset = 0
startTime = 3000 [s]

Well surface pressure

p = 101325 [Pa]
T = 293l.15 [K]
nPorts = 1



StaticPipe (Add new)
nParallel = 1
length = 6.3
diameter = 0.06
height_ab = 6.3

nParallel = 1
length = 0.22
diameter = 0.025
height_ab = -0.22

A pipe that draws water from the well into the pump

Pump inlet initial pressure

$$\text{p\_start} = \text{Atm. pressure} - \rho g \Delta h$$

Atm. pressure = system.p_start

$\rho \sim 1000$ [kg/m3]

$\Delta h = 6.3$ [m]

② Set Media (fluid model) in text view

```
Version:1.0 StartHTML:0000000107 EndHTML:0000046330 StartFragment:0000000471 EndFragment:0000046292
model WellPumpTest3
  …
  Modelica.Blocks.Sources.Ramp ramp1(duration = 180, height = 1, offset = 0, startTime = 3000) annotation( …);   ⑤
  …
  Modelica.Fluid.Pipes.StaticPipe pipe1(                                                                          ④
    redeclare package Medium = Medium, diameter = 0.025, height_ab = -0.22, length = 0.22) annotation( …);
  Modelica.Fluid.Sources.Boundary_pT boundary(                                                                    ③
    redeclare package Medium = Medium, T = 293.15, nPorts = 1, p = 101325) annotation( …);
  Modelica.Blocks.Sources.Constant const(k = 3000) annotation( …);
  Modelica.Fluid.Pipes.StaticPipe pipe2(                                                                          ②
    redeclare package Medium = Medium, diameter = 0.06, height_ab = 6.3, length = 6.3)  annotation( …);
  Modelica.Fluid.Vessels.ClosedVolume volume1(                                                                    ①
    redeclare package Medium = Medium,
    V = 0.001,
    energyDynamics = Modelica.Fluid.Types.Dynamics.FixedInitial,
    massDynamics = Modelica.Fluid.Types.Dynamics.FixedInitial,
    nPorts = 2,
    p_start = system.p_start - 1000 * 6.3 * Modelica.Constants.g_n,
    use_portsData = false)  annotation( …);
  …
```

③ Run a simulation





Start Time = 0 [s]

Stop Time = 6000 [s]

Nº of calculations = 6000

Integration method ida

Non-linear solver options hybrid

Simulation result







Pressure tank

Reference EBARA CORPORATION 25HPO5.25S https://product-standard-pump.ebara.com/product/detail/P030766

Shape



Tank total volume V = 20 L = 0.02 m3

crossArea = 0.1 m2

height = 0.2 m

Inlet		**Outlet**	
height	0.08 m	height	0.08 m
Inside diameter	0.025 m	Inside diameter	0.015 m

$$crossArea = \frac{\pi D^2}{4} = 0.10521 \text{ m2} \sim 0.1 \text{ m2}$$

$$height = \frac{V}{crossArea} = 0.2 \text{ m}$$

Modify OpenTank to make a pressure tank model PressureTank.

Modeling air in a pressure tank



Volume calculation formula

$V_{air} = V_{total} - V$ Air volume

$V_{total} = crossArea \times height$ Total volume

$V = crossArea \times level$ Water volume

Air properties air : Modelica.Media.Air.DryAirNasa.BaseProperties

Air state variables

$M_{air} = air.d \cdot V_{air}$ mass

$U_{air} = air.u \cdot M_{air}$ Internal energy

Air equation

$$\frac{dM_{air}}{dt} = 0$$ Mass conservation

$$\frac{dU_{air}}{dt} = -Wb_{flow}$$ Energy conservation

$$Wb_{flow} = -medium.p \cdot \frac{dV}{dt}$$ Energy from work received by water

Initial condition

$air.p = p\_ambient$ Air pressure

$air.T = T\_ambient$ Air temperature

$level = level\_start$ Water level

$medium.p = p\_ambient$ Water pressure

$medium.T = T\_start = T\_ambient$ Water temperature

Pressure tank inlet / outlet pressure

Consider the hydrostatic pressure in the tank.

$$
\begin{aligned}
p_{vessel} &= vessel\_ps\_static[i] \\
&= \max(0, level - portsData\_height[i]) \cdot system.g \cdot medium.d + air.p \\
&= \rho g \Delta h + air.p
\end{aligned}
$$

usePortsData = true To take into account the pressure loss at the entrance and exit.

Inlet

$$p_{in} + \frac{\dot{m}^2}{2\rho_{in}A_{in}^2} = p_{vessel} + \frac{\dot{m}^2}{2\rho_{vessel}A_{vesssel}^2} + \zeta_{in}\frac{\dot{m}^2}{2\rho_{in}A_{in}^2} \qquad \zeta_{in} = 1.04$$

Static pressure | Dinamic pressure | Static pressure | Dinamic pressure | Pressure loss due to sudden expansion

$$\Rightarrow p_{in} = p_{vessel} + \left(\zeta_{in} - 1 + \frac{A_{in}^2}{A_{vessel}^2}\right)\frac{1}{2\rho_i n A_{in}^2}\dot{m}^2$$

Outlet

$$p_{vessel} + \frac{\dot{m}^2}{2\rho_{vessel}A_{vessel}^2} = p_{out} + \frac{\dot{m}}{2\rho_{out}A_{out}^2} + \zeta_{out}\frac{\dot{m}^2}{2\rho_{out}A_{out}^2} \qquad \zeta_{out} = 0.5$$

Static pressure | Dinamic pressure | Static pressure | Dinamic pressure | Pressure loss due to sudden contraction

$$\Rightarrow p_{out} = p_{vessel} - \left(\zeta_{out} + 1 - \frac{A_{out}^2}{A_{vessel}^2}\right)\frac{1}{2\rho_{vessel}A_{out}^2}\dot{m}^2$$

PressureTank

Model a pressure tank

① Right-click the library browser Modelica.Fluid.Vessels.OpenTank and select Duplicate to create a PressureTank.



Name: PressureTank

Path : ShallowWellPumpingSystem

② Paste Modelica.Blocks.Interfaces.RealOutput in diagram view and name it tankPressure.



③ In text view, add import statements to adjust scope, edit variables, parameters, etc.

```
model PressureTank "Simple tank with inlet/outlet ports"
  import Modelica.Constants.pi;
  import SI = Modelica.SIunits;
  import Modelica.Fluid.Types;
  import Modelica.Fluid;
  // Air
  replaceable package Air = Modelica.Media.Air.DryAirNasa;
  Air.BaseProperties air;
  SI.Volume V_air;
  SI.Mass M_air;
  SI.Energy U_air;
  // Tank properties
  SI.Height level(stateSelect = StateSelect.prefer, start = level_start_eps) "Level height of tank";
  SI.Volume V(stateSelect = StateSelect.never) "Actual tank volume";
  // Tank geometry
  parameter SI.Height height = 0.2 "Height of tank";
  parameter SI.Area crossArea = 0.1 "Area of tank";
  parameter SI.Volume V_total = height * crossArea;
  // Ambient
  parameter Medium.AbsolutePressure p_ambient = system.p_ambient "Tank surface pressure" annotation( ⋯);
  parameter Medium.Temperature T_ambient = system.T_ambient "Tank surface Temperature" annotation( ⋯);
  // Initialization
  parameter SI.Height level_start(min = 0) = 0.08 "Start value of tank level" annotation( ⋯);
```

Adjust the scope

Physical property model of air
State variables

Tank shape

Initial water level

④Set the parameters of the inherited part of the upper model PartialLumpedVessel.

```
  extends Modelica.Fluid.Vessels.BaseClasses.PartialLumpedVessel(
    final fluidVolume = V,
    final fluidLevel = level,
    final fluidLevel_max = height,
    final vesselArea = crossArea,
    heatTransfer(surfaceAreas = {crossArea + 2 * sqrt(crossArea * pi) * level}),
    final initialize_p = false,
    final p_start = p_ambient,
    energyDynamics = Modelica.Fluid.Types.Dynamics.FixedInitial,
    massDynamics = Modelica.Fluid.Types.Dynamics.FixedInitial,
    nPorts = 2,
    portsData = {
      Modelica.Fluid.Vessels.BaseClasses.VesselPortsData(diameter = 0.025, height = 0.08),
      Modelica.Fluid.Vessels.BaseClasses.VesselPortsData(diameter = 0.015, height = 0.08)},
    use_portsData = true
    );
  Modelica.Blocks.Interfaces.RealOutput tankPressure annotation( …);
…
```

Set the initial temperature and pressure to the set values.

Setting the tank inlet

⑤Edit the air and volume equations.

```
equation
// Total quantities
  V = crossArea * level "Volume of fluid";
  medium.p = air.p;
  tankPressure = air.p - p_ambient;
  V_air = V_total - V;
  M_air = air.d * V_air;
  U_air = air.u * M_air;
  der(M_air) = 0;
// Source termsEnergy balance
  if Medium.singleState or energyDynamics == Types.Dynamics.SteadyState then
    Wb_flow = 0 "Mechanical work is neglected, since also neglected in medium model (otherwise unphysical small
temperature change, if tank level changes)";
    der(U_air) = 0;
  else
    Wb_flow = -medium.p * der(V);
    der(U_air) = -Wb_flow;
  end if;
//Determine port properties
  for i in 1:nPorts loop
    vessel_ps_static[i] = max(0, level - portsData_height[i]) * system.g * medium.d + air.p;
  end for;
initial equation
  if massDynamics == Types.Dynamics.FixedInitial then
    level = level_start_eps;
  elseif massDynamics == Types.Dynamics.SteadyStateInitial then
    der(level) = 0;
  end if;
  air.p = p_ambient;
  air.T = T_ambient;
  medium.p = p_ambient;
  annotation( …);
end PressureTank;
```

Boundary conditions
Output to pressure switch
Mass and internal energy
Mass conservation equation

Energy conservation equation

Initial condition

⑥ Switch to icon view and edit the icon.

Output to pressure switch



Draw a rectangle, Right-click and select properties

- Line style
- Fill type

And so on

The outer lines were also deleted and redrawn as rectangles.

Design as you like!

PressureTankTest1 Unit test model

①Right-click on ShallowWellPumpingSystem in the library browser and select Create New Modelica Class. Create model PressureTankTest1 and arrange and edit as follows.

Modelica.Blocks.Source.Ramp
height = -0.2
duration = 10
offset = 0
startTime = 120

Modelica.Blocks.Source.Ramp
height = 0.1
duration = 10
offset = 0
startTime = 0



Modelica.Fluid.Sources.MassFlowSource_T
use_m_flow_in = true
nPorts = 1

② Switch to text view and set Media (fluid property model).

```
model PressureTankTest1
  replaceable package Medium = Modelica.Media.Water.StandardWater;
  PressureTank tank(redeclare package Medium = Medium) annotation( ⋯);
  Modelica.Fluid.Sources.MassFlowSource_T boundary(
    redeclare package Medium = Medium, nPorts = 1, use_m_flow_in = true)  annotation( ⋯);
  Modelica.Fluid.Sources.MassFlowSource_T boundary1(
    redeclare package Medium = Medium, m_flow = 0, nPorts = 1, use_m_flow_in = true)  annotation( ⋯);
  Modelica.Blocks.Sources.Ramp ramp1(duration = 10, height = 0.1, offset = 0, startTime = 0)  annotation( ⋯);
  inner Modelica.Fluid.System system annotation( ⋯);
  Modelica.Blocks.Sources.Ramp ramp2(duration = 10, height = -0.2, offset = 0, startTime = 120) annotation( ⋯);
equation
  connect(ramp2.y, boundary1.m_flow_in) annotation( ⋯);
  connect(ramp1.y, boundary.m_flow_in) annotation( ⋯);
  connect(tank.ports[2], boundary1.ports[1]) annotation( ⋯);
  connect(boundary.ports[1], tank.ports[1]) annotation( ⋯);
  annotation( ⋯);
end PressureTankTest1;
```

③ Run a simulation





Start Time = 0 [s]

Stop Time = 250 [s]

Number of calculations = 500

Integral method ida

Non-linear solver option hybrid

Simulation results









It was confirmed that when the water level rose, the air was compressed and the tank pressure increased.

PressureTankTest2



① **Create PressureTankTest2 by duplicating WellPumpTest3.**

② **In the text view, find OpenTank, replace it with PressureTank, Delete a meter.**

```
Modelica.Fluid.Vessels.OpenTank tank(redeclare package Medium = Medium, crossArea = 0.01, …
```

```
ShallowWellPumpingSystem.PressureTank tank(redeclare package Medium = Medium) annotation( …);
```

③ **Edit the parameters of ramp1.**

height = 1
duration = 10 [s]
offset = 0
startTime = 35 [s]



④ Run simulation





Start Time = 0 [s]

Stop Time = 140 [s]

Number of calculations = 500
Integral method ida
Non-linear solver option hybrid

Simulation results









PressureTankTest2



PressureSwitch

① Create PressureSwitch by duplicating PressureTank2 and edit as follows.

Modelica.Blocks.Logical.Hysterisis
uLow = 8e4 [PaG] p_{on}
uHigh = 2.5e5 [PaG] p_{off}
pre_y_start = false

Modelica.Blocks.Logical.Not

Modelica.Blocks.Logical.TriggeredTrapezoid
amplitude = 3000 [rpm]
rising = 1 [s]
falling = 1 [s]
offset = 0 [rpm]
pre_y_start = false



Modelica.Blocks.Sources.Trapezoid
amplitude = 1
rising = 5 [s]
width = 30 [s]
falling = 5 [s]
period = 120 [s]
nperiod = -1
offset = 0
startTime = 60 [s]

The pressure at which the pump turns off

$$p_{off} = \rho g(h_{max} - h_{well}) = 2.54223 \times 10^5 \sim 2.5 \times 10^5 \text{ [PaG]}$$

The pressure at which the pump turns on

$$p_{on} = 0.8 \times 10^5 \text{ [PaG]}$$

ρ	density(25℃)	997.062	kg/m3
g	Gravitational acceleration	9.80665	m/s2
h_{max}	Maximum head	32	m
h_{well}	Well depth	6	m

The operation of the pressure switch after execution will be described.

② Run a simulation





Start Time = 0 [s]

Stop Time = 1000 [s]

Number of calculations = 1000

Integral method ida

Non-linear solver option hybrid

Tank pressure [PaG]



Operation of pressure switch

hysteresis1



Pressure tank
On at 2.5 × 105 [PaG] or more
Off at 0.8 × 105 [PaG] or less

not1



Signal inversion
Off when 2.5 × 105 [PaG] or more
On at 0.8 × 105 [PaG] or less

triggerdTrapezoid1 [rpm]



Converts on / off signal to pump speed
Rising 1 s
Falling 1s

Pump volume flow [m3/s]



Simulation result

Valve opening



Flow rate [m3/s]



Water level in pressure tank [m]



Tank pressure [PaG]



Pump speed [PaG]



Pump head [m]



Conclusion

• For stop valves, pumps, well water pipes, pressure tanks and pressure switches
A model was created and a simple shallow well pump water supply system was modeled.

• Simulate the case where the stop valve opens and closes periodically to determine the flow rate and Changes in pressure, tank water level, etc. were obtained.

Next deployment

• Perform more realistic modeling of the rising and falling times of the rotation speed when the pump is turned on and off.

• Assuming a kitchen, wash basin, bath, toilet, etc., model by placing multiple stop valves.